
Even though the ACL2 logic is first order, the ACL2 system offers several mechanisms provid-
ing users with some operations akin to higher order logic ones. In this paper, we propose a macro,
namedinstance-of-defspec, to ease the reuse of abstract functions and facts proven about them.
Defspec is an ACL2 book allowing users to define constrained functions and their associated prop-
erties. It contains macros facilitating the definition of such abstract specifications and instances
thereof. Currently, lemmas and theorems derived from theseabstract functions are not automatically
instantiated. This is exactly the purpose of our new macro.instance-of-defspec will not only
instantiate functions and theorems within a specification but also many more functions and theorems
built on top of the specification. As a working example, we describe various fold functions over
monoids, which we gradually built from arbitrary functions.

1 Introduction

The primary goal of this paper is to provide a way to reason abstractly in ACL2, while being able to
specialise later. This kind of reasoning is necessary in thedevelopment of large modular proofs. Our
main contribution is a macro, calledinstance-of-defspec, providing a convenient way to manipulate
abstract functions and their properties. In our approach, we build generic theorems, and use our macro
to apply them on more specific instances, that is: reuse proofs. For this to work, we reuse functions
as well, by instantiating generic functions with more specific ones. While both function- and proof
reuse are possible to some extent in ACL2, we provide a singlemacro to do both in a way that is more
convenient than the existing solutions. Aside from the theoretical example presented in our paper, we
briefly discuss the use of our solution in the development of ageneric theory of communication networks,
called GENOC [10]. This effort is a large and modular proof development of about fifty thousands lines
of ACL2 code. Our macro already provides a thousand lines reduction of our code base.

This paper uses monoid-operations and fold operations as leading examples. A monoid is simply a
closed and associative operation with an identity element,for which we will write◦ and 0 respectively. A
fold operation is an operation that changes a list(a0 . . .ak) into a value(a0◦· · ·◦ak). The leading example
is not of particular importance on its own, but used to illustrate the macroinstance-of-defspec.

Using theencapsulate environment, ACL2 provides a way to hide function definitions, while
preserving certain function properties. We can then prove atheorem about such functions, using only the
function properties in the proof. If we would then write a newfunction that satisfies all the properties
used in the proof, we know that the function will satisfy the theorem as well. However, even though we
know that the function satisfies the theorem, the only way to actually use this knowledge in further proofs
is by telling it to ACL2 in the form of a new theorem. Using an uninterpreted function in the definition
of a concrete function raises similar issues. We provide a macro calledinstance-of-defspec, that
will allow ACL2 users to instantiate abstract functions while being able to use higher order theorems
and functions.

The nameinstance-of-defspec derives from the worddefspec, which already is a part of
ACL2 [9]. In essence,defspec is a labeledencapsulate environment, such that the functions which
are kept local to theencapsulate can be referred to as a single package. The advantage ofdefspec

is that other functions can be declared to be an instance of this defspec. The current way of doing this
in ACL2 is by usingDefInstance, which is a part ofdefspec. This enables us to prove the ‘higher
order’ statement that addition is a monoid. One could then use the:functional-instance hint, which
was already implemented in Nqthm[1], to prove a specific theorem if it was already proven in a more
general context. For example, we can use it to prove a theoremfor addition which was already proven
for a monoid efficiently.

Our approach combines these previous solutions. At a call ofinstance-of-defspec, the logical
world is searched for functions and theorems that depend on the abstractdefspec. These are copied for
the concrete instance. UsingDefInstance and the:functional-instance hint, the proof of every
theorem is repeated for the instance without the computational burden.

Sinceinstance-of-defspecwill copy functions, our approach offers an alternative todefattach.
Kaufmann and Moore [5] explain how this can be used to add an executable function, to for instance the
abstract monoid, such that we can execute it. Unfortunately, defattach only allows one such attach-
ment per function, and it does not influence the logical world. Our macro can be seen as an extension
to DefInstance which combines all of these solutions, making a copy of functions instead of using
defattach thus actually applying the change in the logical world.

As a leading example for reasoning abstractly, we develop a small theory about monoids. We prove
that (x0 ◦ (x1 ◦ (. . . ◦ (xn ◦0)))) = (((x0 ◦ x1) ◦ . . .) ◦ xn) for n≥ 0, using any associative operation for◦

with identity element 0. The benefit of such a proof is that we can apply it to arithmetic addition with
zero, to multiplication with one, or to appending lists withthe empty list.

In this paper, we present this example explaining howinstance-of-defspec can be used in Sec-
tion 2. In Section 3 we look at the inner workings of the macro,explaining limitations and opportunities
in our approach. We also describe howinstance-of-defspec can be used to add extra arguments to a
function. Similar approaches are compared toinstance-of-defspec in Section 4

2 Monoids as an example usage

Definition 1 (Monoid) A monoid(0,◦) on the domain D is an operator◦ and a constant0 such that:

• ◦ is closed: for all a,b∈ D we have(a◦b) ∈ D.

• ◦ is associative:(a◦b)◦c= a◦ (b◦c) for a,b,c∈ D.

• 0∈ D is an identity element:0◦a= a= a◦0.

In this section, we build a monoid by adding the three constraints from the definition one at a time. On
the unrestricted operator◦, we define different implementations of a fold operation. Wewill show that
they are equivalent for monoids.

This section is self-containing, that is: the ACL2 interpreter should accept the inputs given in this
section as-is. For additional theorems and examples, or forless typing, the reader may use the file
closedMonoid.lisp.

First, we include our instance-of-defspec book.

1 (include-book "instance-of-defspec")



S.J.C. Joosten, B.E. van Gastel, J. Schmaltz 31

This book includes an entirely abstract function calledbinary-function, which was encapsulated in a
defspec calledbinary. We repeat the definition ofbinary here. Note that this event cannot be entered
in ACL2 here, sincebinary is already present in our instance-of-defspec book.

(defspec binary ((binary-function (x y) t))

(local (defun binary-function (x y) (cons x y))))

Here we usedefspec, together withlocal. The actual implementation of alocal event is unknown to
ACL2 outside thedefspec. The only thing ACL2 knows aboutbinary-function is that it takes two
arguments. Hence we can regardbinary-function as an arbitrary binary function, despite the chosen
implementation ofcons here.

2.1 Reusing functions

The intuition of the fold operation is that it transforms a list (x1 . . .xn) into a valuex1 ◦ · · · ◦ xn for any
binary operator◦. For transforming the empty list, we need some sort of identity element to build upon.
For this reason, we provide a first elementx0. We definefoldl andfoldr, which differ in the placement
of the brackets. In the case offoldl, the brackets are written as((((x0 ◦x1)◦x2)◦x3)◦ · · ·). In the case
of foldr andfoldr1, brackets are written as:x1◦ (x2◦ (x3◦ (· · · ◦x0))). Forfoldr1 we require that the
list has at least one element. Forfoldr (foldl) we supply the last (first) element.

Note that, in the end, we will prove that these fold operations all are equivalent, even when providing
an identity element as the first element. For this proof, we need associativity and an identity element,
which is precisely the requirement of a monoid.

2 (defun foldr (x xs)

3 (if (atom xs) x (binary-function (car xs) (foldr x (cdr xs)))))

4

5 ; Alternatively, we can ’omit’ the first element:

6 (defun foldr1 (xs)

7 (if (atom (cdr xs)) (car xs)

8 (binary-function (car xs) (foldr1 (cdr xs)))))

9

10 (defun foldl (x xs)

11 (if (atom xs) x

12 (foldl (binary-function x (car xs)) (cdr xs))))

Since the focus of this article is the use ofinstance-of-defspec, we do not proceed by proving
properties of the fold functions, but show how to actually use these fold functions. A trivial example of
a binary function iscons. We can instantiatebinary-function (provided by thedefspec encapsulate
binary) with cons under a list of substitutions as follows:

13 (instance-of-defspec binary cons ’((binary-function cons) (foldr cons-foldr)

14 (foldr1 cons-foldr1) (foldl cons-foldl)))

This instantiates our fold functions as executable functions:

ACL2 !> (cons-foldr ’a ’(b c))

(B C . A)

ACL2 !> (cons-foldr1 ’(a b c))

(A B . C)

ACL2 !> (cons-foldl ’a ’(b c))

((A . B) . C)

Now lets add the assumption that our binary function is closed. That is: there is some domain, and if
both arguments to the binary function belong to it, so does its result.

We encapsulate it using adefspec, and instantiate it as a binary operation. We will explain the use
of defspec in Section 3.1. You may think of it as anencapsulate environment that hides the local
definitions and provides the notion ofclosed-binop in terms ofc-domainp, c-binary-function
and the theorem (to be seen as a property)closed-binop-closed.

15 (defspec closed-binop ((c-domainp (x) t)

16 (c-binary-function (x y) t))

17 (local (defun c-domainp (x) (integerp x)))

18 (local (defun c-binary-function (x y) (+ x y)))

19 (defthm closed-binop-closed

20 (implies (and (c-domainp x) (c-domainp y))

21 (c-domainp (c-binary-function x y)))))

22 (instance-of-defspec binary c) ; choose c (for closed) as the prefix symbol here

In the last statement, we instantiateclosed-binop as an arbitrary binary operator: we use abstract
functions as the instantiation of other abstract functions. Even though a closed binary function is an
abstraction itself, it is an instantiation of the - more general - binary function.

Note that in this case, we did not specify the list of replacements. We do not have to: the sec-
ond argumentc is used as the default prefix. Hence,c-foldr is the instantiation offoldr with
c-binary-function asbinary-function:

ACL2 !> :pf c-foldr

(EQUAL (C-FOLDR X XS)

(IF (CONSP XS)

(C-BINARY-FUNCTION (CAR XS)

(C-FOLDR X (CDR XS)))

X))

In fact, we could have done the same withcons. The following would have been a shorter notation for
the same instruction we gave earlier:

23 (instance-of-defspec binary cons ’((binary-function cons)))

2.2 Reusing theorems

We have seen how to reuse functions usingdefspec. We can do the same trick for theorems. In the
context of a closed operation, we show that the repetitive application of this operation again yields an
element in its domain. For brevity, we only show this for thefoldr1 operation:

24 (defun list-domainp (xs)

25 (if (endp xs) t

26 (and (c-domainp (car xs)) (list-domainp (cdr xs)))))

27

28 (defthm foldr1-closed

29 (implies (and (list-domainp xs) (consp xs))

30 (c-domainp (c-foldr1 xs))))

We reuse this theorem for a semigroup. A closed associative operator◦ is called a semigroup. Note
that a semigroup is like a monoid, but without the identity element 0. More formally: a monoid(0,◦)
is a semigroup◦ with identity element 0. So once again, we write adefspec, and specify that it is an
instance of a closed binary operator.

31 (defspec semigroup ((sg-c-domainp (x) t)

32 (sg-c-binary-function (x y) t))

33 (local (defun sg-c-domainp (x) (integerp x)))

34 (local (defun sg-c-binary-function (x y) (+ x y)))

35 (is-a closed-binop sg semigroup-is-a-closed-binop)

36 (defthm semigroup-assoc

37 (implies (and (sg-c-domainp x)

38 (sg-c-domainp y)

39 (sg-c-domainp z))

40 (equal (sg-c-binary-function x (sg-c-binary-function y z))

41 (sg-c-binary-function (sg-c-binary-function x y) z)))))

42 (instance-of-defspec closed-binop sg) ; reuse the fold operators (again)

Note that we used the macrois-a inside our semigroup. This copies the theorems from the closed
binary operator into the currentdefspec. By doing so, we ensure that the previously defined specifica-
tion of closed-binop will be copied tosemigroup, since this is required to prove that asemigroup is
an instance ofclosed-binop. Theis-a macro is auxiliary toinstance-of-defspec. Its implemen-
tation will be discussed in Section 3.5. In this case, the following theorem is generated in place of the
is-a macro:

ACL2 !> (OLDSPEC ’CLOSED-BINOP ’SEMIGROUP-IS-A-CLOSED-BINOP ’SG () (W STATE))

((DEFTHM SEMIGROUP-IS-A-CLOSED-BINOP-0

(IMPLIES (IF (SG-C-DOMAINP X)

(SG-C-DOMAINP Y)

’NIL)

(SG-C-DOMAINP (SG-C-BINARY-FUNCTION X Y)))))

Now we try to prove thatfoldr1 is foldl. Note that this was not the case forcons-foldr1 and
cons-foldl. We need to prove it for the semigroup folds:sg-c-foldr1 andsg-c-foldl.

43 (defthm foldr1-is-foldl

44 (implies (and (sg-c-domainp x) (sg-c-domainp y)

45 (sg-list-domainp xs))

46 (equal (sg-c-foldr1 (cons x xs))

47 (sg-c-foldl x xs))))

If we look at the proof output, we will find that the proof has used a theorem we did not define
ourselves, but which was automatically copied based on the theorem we added to the closed binary
operator:

ACL2 !> :pf (:REWRITE SG-FOLDR1-CLOSED)

(IMPLIES (AND (SG-LIST-DOMAINP XS) (CONSP XS))

(SG-C-DOMAINP (SG-C-FOLDR1 XS)))

Without the theoremfoldr1-closed, and thus the automatically derived theoremsg-foldr1-closed,
the proof attempt offoldr1-is-foldl would have failed.

2.3 On monoids

As promised, we end with a theory about monoids. A monoid is a semigroup with an identity element.
To take care of the names, we use a renaming constant. Note that we can use renaming in the macro
is-a, exactly like ininstance-of-defspec.

48 (defconst *monoid-renaming*

49 ’((sg-c-domainp mon-domainp) (sg-c-foldr mon-foldr)

50 (sg-c-binary-function mon-binop) (sg-c-foldr1 mon-foldr1)

51 (sg-list-domainp mon-list-domainp) (sg-c-foldl mon-foldl) ))

52

53 (defspec monoid ((mon-domainp (x) t) (mon-binop (x y) t)

54 (mon-id () t))

55 (local (defun mon-domainp (x) (integerp x)))

56 (local (defun mon-binop (x y) (+ x y)))

57 (local (defun mon-id () 0))

58 (defthm id-in-domain (mon-domainp (mon-id)))

59 (is-a semigroup mon monoid-is-a-semigroup *monoid-renaming*)

60 (defthm monoid-id-left

61 (implies (and (mon-domainp x))

62 (equal (mon-binop x (mon-id))

63 x)))

64 (defthm monoid-id-right

65 (implies (and (mon-domainp x))

66 (equal (mon-binop (mon-id) x)

67 x))))

68 (instance-of-defspec semigroup mon *monoid-renaming*)

We introduce functionfold which acts likefoldr1, but without the requirement that its argument
should be aconsp.

69 (defun fold (xs) (if (atom xs) (mon-id) (mon-foldr1 xs)))

We end the book about monoids, the fileclosedMonoid.lisp, by proving equality between the different
versions offold, and giving another instantiation. These proofs are omitted here, but we encourage the
reader to take a look.

3 Inner workings

As the sources will be made available with this publication,we do not reproduce them here. Instead,
we highlight the main parts to give the reader a rough understanding of the code, and highlight the ‘ugly
bits’ to illustrate the difficulties and limitations of our approach.

The macroinstance-of-defspec effectively does three things:

1. Look up thedefspec and prove that the provided instance is an instance of thisdefspec using
DefInstance.

2. Look up every function that uses one of thedefspec functions, and copy it as an instantiated
function.

3. Look up every theorem that use one of the functions from theprevious step, and copy it as an
instantiation.

In order to find the functions and theorems, we look at the ACL2world. For this reason, we use
make-event. In particular,instance-of-defspec is a macro that expands to:

‘(make-event (instanceOf-defspec-fn ’,spec ’,prefix ,rename state)

:check-expansion nil)

3.1 Obtaining thedefspec

To reason abstractly, ACL2 provides theencapsulate environment. This environment allows the user
to hide events, such that a particular theorem can be stated about a function, without allowing the defini-
tion of this function to enter the logical world. When proving further properties outside the encapsulate,
the function is seen as an abstract function, since the function is unknown to the logical world.

To prove that a certain concrete function is an instance of this environment,defspec was developed
by Sandip Ray and Matt Kaufmann. The system bookmake-event/defspec.lisp provides two parts:
the first is a macro calleddefspec, which does exactly the same as anencapsulate environment, but
then also provides a name for it. The second is a macro calleddefinstance, which generates a theorem
equivalent to stating that some implementation is an actualinstantiation of thedefspec named. We
chose to build on this approach for the reason why Ray and Kaufmann developed it. In the comments of
their code, they write [9]:

The real problem is that ACL2 is a theorem prover for first order logic, not higher-order
logic, while the statement we want to make is inherently a higher-order statement. (...)

But having one macro that generates the instances will give the evaluators the ability to trust
it, rather than hand-coded proofs that all the “corresponding” constraints are satisfied. The
hope is that with a lot of use the macros here will be conventionally thought of as higher-
order representations.

To obtain thedefspec, we useacl2::decode-logical-name to skip to the place where the def-
spec was declared, and lookup the corresponding encapsulate event. In the case ofclosed-binop as
written in the previous section, it might look like this:

(EVENT-LANDMARK GLOBAL-VALUE 8844

(ENCAPSULATE (C-DOMAINP C-BINARY-FUNCTION)

. :COMMON-LISP-COMPLIANT)

ENCAPSULATE

((C-DOMAINP (X) T)

(C-BINARY-FUNCTION (X Y) T))

(LOCAL (DEFUN C-DOMAINP (X) (INTEGERP X)))

(LOCAL (DEFUN C-BINARY-FUNCTION (X Y) (+ X Y)))

(DEFTHM CLOSED-BINOP-CLOSED

(IMPLIES (AND (C-DOMAINP X) (C-DOMAINP Y))

(C-DOMAINP (C-BINARY-FUNCTION X Y)))))

From this, we obtain the corresponding functions by taking thecadar of the third element. In this case,
it returns(C-DOMAINP C-BINARY-FUNCTION). At this point, we would like to note that we do not know
whether this is an appropriate way to find the encapsulated functions. We just chose this to identify the
defspec because it consistently returned these functions in multiple tests.

Instantiating thedefspec is done with adefinstance. At the instantiation ofclosed-binop as a
semigroup, we generated the following statement:

(DEFINSTANCE CLOSED-BINOP SG-CLOSED-BINOP

:FUNCTIONAL-SUBSTITUTION ((C-DOMAINP SG-C-DOMAINP)

(C-BINARY-FUNCTION SG-C-BINARY-FUNCTION)))

This is the first statement generated. It is also the main proof obligation: it will fail when ACL2 cannot
prove that the instance presented is an implementation of thedefspec.

3.2 Obtaining functions

Once we have found the functions defined in thedefspec, we look up all functions that depend on these.
Note that we apply this transitively. That is: if we use the abstract functionf insideg, andg is used inside
h, thenh has to be instantiated as well. To do so, we wrote functionget-derived-funs.

Functionget-derived-funs goes through theworld multiple times, keeping track of a list of
‘discovered’ functions while looking for functions that use one of these and adding them to the list. The
search for functions is done by looking forDEF-BODIES in theworld, which ensures that macros have
been eliminated from the definition. We terminate this search once the list does not grow any further.
For each separate function, we stop searching once we hit thedefinition of that function. The rationale
behind this is that a function cannot be used before it was declared. Since the procedure will just be
used in amake-event, we can leave it in:program mode, saving us the trouble to prove termination.
We take care not to add functions twice, which does provide termination: there are a finite number of
functions in the current logical world.

When copying the function, we wish to be as similar to the original function as possible. However,
we have to replace the abstract functions for their instantiations. To do so, we expand all macros in the
function body using the:trans macro. After this, we replace the functions with their definitions using
a function we calledreplacefns. This function takes a list of desired substitutions, and a list of terms
in which this replacement should take place.

ACL2 !> :replacefns ((foo bar) (bar foo))

((+ ((lambda (foo j) (foo foo j)) x y) (bar x y)))

((+ ((LAMBDA (FOO J) (BAR FOO J)) X Y) (FOO X Y)))

Copying our function as identically as possible has the advantage of copying documentation and
other parameters as well. There are cases where the generated defun fails, which occur when ACL2
fails to prove guards or termination of the generated function.

3.3 Obtaining theorems

Obtaining the theorems happens in a similar way. The main difference is that the list of functions does
not grow while searching for theorems that use these functions. We can therefore find all theorems in
one pass through theworld.

When generating the theorems, we do not try to let the copied theorem mimic the original, as we did
with functions. In the case of functions, it was possible that ACL2 tried to prove something (like guards

or termination), but failed. In the case of theorems, we wantto prevent this from happening, by making
sure that ACL2 does not redo an entire proof. To achieve this,we use the:functional-instance
hint. Also, we do not try to copy the original event, but look for its effect on the logical world and copy
the effect.

The theoremfoldr1-closed from Section 2.2 is stored in theworld as:

(...

(FOLDR1-CLOSED THEOREM IMPLIES

(IF (LIST-DOMAINP XS) (CONSP XS) ’NIL)

(C-DOMAINP (C-FOLDR1 XS))) ...

(FOLDR1-CLOSED CLASSES (:REWRITE))

...)

Note that the theorem and the classes are stored in differentworld items. We obtain the theorems by
combining these two parts in theworld. The first holds the (translated, macro-free) theorem, and the
second holds the rule-class(es). Some rule-classes are stored with extra attributes. For example, a typing
rule will define a typed term, and a forward-chaining rule hasa trigger-term.

The :functional-instance hint will generate various subgoals, which can be proven using the
proof generated by a previously provendefinstance theorem (which is the first theorem we generate,
even before copying the functions). We also add the:in-theory hint, giving a theory that contains only
the newly generated functions. Although we do not know how toprove that this is a sound way to copy
theorems, we can at least give the anecdotical evidence thatwe have not found an example where the
proof attempt for a generated theorem failed.

While developinginstance-of-defspec, we found it useful to be able to keep track of the theo-
rems defined. With relatively little effort, we have writtenthe macrosymbol-lemmas which, like:pl,
shows theorems containing its argument. The main difference being that the former shows all theorems,
while the latter only shows those in which its argument is a trigger.

3.4 Adding arguments

The main use ofinstance-of-defspec is to provide a single method to create functions likemap, and
corresponding theorems. An issue we did not foresee was thatsome of the functions we would like to
map, though effectively unary, are actually binary.

We have solved this by allowing for lambda functions in the function substitution. As an example,
we investigate a generic function that checks whether all elements in some list satisfy some predicate:

(defspec list-predicate ((predicate (x) t))

(local (defun predicate (x) x)))

(defun predicate-listp (lst)

(if (atom lst) (null lst) ; require true-lists

(and (predicate (car lst)) (predicate-listp (cdr lst)))))

Now suppose we want our predicate to bemember-equal. This would enable us to test whether all
elements in some list occur in some other list, which is exactly the condition for subsets. We can create a
new functionsubset-equal (written without ap to prevent collision with the builtinsubsetp-equal):

(instance-of-defspec list-predicate members

’((predicate (lambda (x) (member-equal x y)))

(predicate-listp (lambda (lst) (subset-equal lst y)))))

At this point, we require the lambda expression to have exactly the same argument as the arguments of
the function it replaces:

ACL2 !> (instance-of-defspec list-predicate members

’((predicate (lambda (x) (member-equal x y)))

(predicate-listp (lambda (xs) (subset-equal xs y)))))

ACL2 Error in COPYFUN: The lambda construct

(LAMBDA (XS) (SUBSET-EQUAL XS Y)) takes as input (XS), which should

be an exact match of the original arguments of the original function:

(LST)

This syntactic limitation will hopefully prevent users unintentionally swapping arguments, and helps
identify which variables are substituted.

3.5 Theis-a macro

To prove something is an instance of somedefspec, we need to prove all theorems included in that
defspec. If we want to add a property to a previousdefspec, we proceed by creating a newdefspec,
and indicating that the new is an instance of the former. One way to do this, is by usingdefinstance,
and another is by usingis-a. The main advantage of the latter is that theorems are be copied individually,
and can remain enabled. This results in a different behaviour of subsequent proof attempts. Another
advantage is that the notation ofis-a is very similar to that ofinstance-of-defspec.

The implementation ofis-a is a lot like definstance, in the sense that it uses the function
constraint defined in thedefspec book to get all theorems that have to hold. For every func-
tion it finds in thedefspec, the functionconstraint is called which returns a theorem that must
hold. While it would be possible to useis-a outside of adefspec, it would make more sense to use
instance-of-defspec there.

4 Discussion

Related work Before we started working withinstance-of-defspec, we used ACL2’s macro sys-
tem to avoid code duplication. This approach does not involve inspecting the world, ormake-event.
We just used plain macros that write out ‘instances’. Apart from functions, we used these macro’s to
generate theorems as well. In this approach we lose the proofobligation that the proposed instance is
actually an instance, or even having to create an encapsulated environment for the abstract structure. The
price we pay, however, is having to do full proofs for all instantiated theorems, which can be a heavy
computational burden. Also, we found the current approach of writing theorems about abstract instances
directly in the logical world to be more convenient than writing theorems in a macro. Carl Eastlund and
Matthias Felleisen make a case against using ACL2’s macros,proposing hygienic macros [2]. If we
were to use hygienic macros for this purpose, writing out functions and theorems using a macro could be
more convenient. We would still need to redo theorems.

A very promising alternative to our approach was presented by Gamboa and Patterson [4]. The
authors introduce a way to use polymorphism in ACL2. One of the downsides is the use of astobj
calledmemory. In addition, it is presented as an alternative to the current encapsulate environments,
instead of building on what is already present in ACL2.

Although we could not find a paper on it, there is a macro calleddef-functional-instance

inside the ACL2 tools directory which provides an easier wayto instantiate previous theorems using
the:functional-instance hint. The problem is solved more rigorously by Moore [8], by providing
an alternative to the:functional-instance hint which computes its substitutions automatically. Our
instance-of-defspec macro does all this, and these two previous solutions stressthe importance of
doing so.

For conveniently instantiating functions, the macro by Martı́n-Mateos et al. [7] is most similar to
ours. The same instantiations are performed: both functions and theories can be reused. The authors did
not use thedefspec environment, nor do their macros rewrite proof hints to aid the instantiated proofs.
Also, at some point in the process the user is required to specify exactly which events to copy (these must
bedefun anddefthm events). A macro call must be placed around these events, andthe output of this
macro should be copied manually. The user then obtains a program which will define a constant that can
be used for reuse. One of the reasons that their approach was less convenient than ours, is that ACL2 did
not have some of its current features, especiallymake-event, at the time. Indeed, such improvements
are present in later work. In the code accompanying the more recent paper from this group [6], many of
the improvements suggested in the earlier paper [7] have been made. Fundamental differences still are:

• The macro by Martı́n-Mateos only copies a handful of events,where our macro will take any event
that rewrites itself to adefun or adefthm, which includes those handled by Martı́n-Mateos.

• The macro by Martı́n-Mateos requires the general theory to be wrapped inside a macro entirely.
Our approach only requires this for the localdefspec, allowing the general theory to be distributed
over multiple books.

• Our macro allows arguments to be added to function calls without changing the original function.

• Leaving the definitions of the instantiation enabled might cause the macro by Martı́n-Mateos to
fail. In our approach the theorem prover is guided by automatically generated hints. Since both
approaches use the:functional-instance hint, we expect this to be something which can be
resolved rather easily.

Carl Eastlund and Matthias Felleisen built a module system on top of ACL2, using the Dracula
environment [3]. In this system, the abstract entities are called interfaces, which have signatures and
contracts. This compares well to thedefspec which is already built in ACL2. A major difference is
that, in ACL2, a function has to be declaredlocal, and the theorems must hold for it. In essence, the
user must provide a witness for thedefspec, while the Dracula environment does not require this. The
interfaces can be instantiated as modules, or reused in modules, by using export and import commands
respectively. An advantage of Dracula Modular ACL is that modules are checked independently, which
might improve the performance of ACL2 by reducing memory requirements. For our approach, one
would have to write different files in order to get such behavior. We see two disadvantages to the modular
approach.

• Reasoning about interfaces directly is not possible. A familiar proof states that there can only be
one identity element for a monoid. In the modular approach, one would have to create a module
which imports the interface, and exports another interfacewhich also has the property that there
is only one identity element. In their paper, Eastlund and Felleisen prove correctness of the exe-
cutable modules. Hence the module that describes a propertyabout the monoid interface will not
be proven to be correct until a witness for the monoid is provided.

• The Dracula environment requires the user to trust both the ACL2 system and the module system,
whereas in our approach all proofs are performed within the ACL2 system. Although we believe

the module system to be sound, bugs in ACL2 up to version 2.3 show that reasoning with encap-
sulated properties is particularly error-prone. The reader may, for instance, read the discussion on
subversive recursions in the miscellaneous section of the ACL2 documentation (version 3 and up).

Use in practice Our group is involved in a large scale generic proof about Network-On-Chips, called
GENOC [10]. The proofs consists of almost fifty thousand lines of ACL2 code. In GENOC all kinds of
properties about classes of networks are proven. To maximise proof reusability, an important aspect is
partially instantiating these generic proofs for classes of networks, and later fully instantiating the proofs
for concrete instantiations of networks. Currently we use many hand coded functional substitution rules
in thedefinstance. These are hard and error-prone to maintain. By using theinstance-of-defspec

construct we already reduce our effort to maintain our codebase. Also a lot of instantiations of generic
theorems can be removed, because they are automatically generated. Both these changes reduced our
code base by one thousand lines.

As instance-of-defspec copies every function and theorem based on adefspec. Because of
this, instantiating a defspec early in a proof session and instantiating it late in the session could have
different meanings. For this reason, we ensured that it would be possible to instantiate the same set of
functions as a defspec twice. One might worry that all extra instantiated definitions can clutter up the
logical world and namespace. In practice, however, we only needed to instantiate each set of functions
as one defspec, and did not run into problems with a clutteredup namespace. Nevertheless we provided
a feature, per request by one of the reviewers, which allows you to not copy some theorems. To do this,
add a theorem to the rename list without providing a new name for it. For instance, ifsometheorem
would normally be instantiated by a call toinstance-of-defspec, add(sometheorem newname) to
usenewname as the new name for this theorem, or(sometheorem) to not copysometheorem at all.

Future work Using our macro inall instances in GeNoC still presents a problem. We cannot handle
function definitions insideencapsulate environments correctly. To find functions to use aslocal

witness, we need to look inside theencapsulate environment, which is forbidden. For this reason,
thedefun-sk anddefevaluator events, which rewrite to events including theencapsulate, are not
supported.

In the current example, we useis-a to include a previousdefspec into a new one. In practice, not
all functions may be duplicated in this step.

We can already illustrate how this problem arises by a modification in the monoid example. Sup-
pose that in Section 2.1, instead of adding thedomainp function toclosed-binop, we would have
usedpredicate and the corresponding generic functionpredicate-listp (possibly included from a
different file). Proving thatclosed-binop is an instance ofbinary will not be a problem. Proving
that thesemigroup is an instance ofclosed-binop, however, requires instantiatingpredicate with
sg-c-domainp while instantiating the semigroup operator withclosed-binop. We are in the process
of finding a convenient syntax for this, and we will provide a way to do this in the near future.

5 Conclusions

The macroinstance-of-defspec enables us to reuse abstract functions and facts proven about them.
It automatically instantiates lemmas and theorems derivedfrom the abstract functions. We are convinced
that together withdefspec, our macro provides a way to reason abstractly in ACL2, whilebeing able
to specialise later. The great variety of similar approaches shows that there is a strong need for abstract

reasoning in ACL2. This variety of solutions also suggests that no approach has become the standard
for the community. We believe that our approach is more convenient than the existing solutions, and we
hope it will become a standard in the ACL2 community.

